

This paper by Walt Lounsbery of Wichita, Kansas provides a very important tool to the analyst who is using the available low Reynolds number wind tunnel data to predict the performance of R. C. sailplane designs. It also shows the effectiveness of "home type" computers in performing the repetative mathematical operations necessary when you're doing this type of study.

Knowledge is one. Its division into subjects is a concession to human weakness.

—HALFORD JOHN MACKINDER

Simple Calculation of Airfoil Moment Coefficients

Walter Lounsbery

"I cannot hope to grasp the impact of this momentous occasion."
- Casey Jones

"What follows is magic."
- Harry Houdini

After reading Dieter Althaus' recent book on model aircraft-type airfoil sections, I was struck by an immediate desire to see how these could be used to improve performance of my models. Many other people seem to have this near-universal reaction, some of which have never attempted any kind of aircraft design before. One cannot help being overcome by such a large collection of reliable experimental data. It is unfortunate that this encyclopedia of section properties does not provide all the information we need to estimate glider performance from airfoil section data.

The stark reality of the situation, at least for performance estimation, is the lack of moment coefficient data on every one of the airfoils in Althaus' book. The lack of moment data is serious, but it is certainly not an intentional omission. There can be no doubt that the people who produced the airfoil data were concerned with the moments the airfoils generated, but the equipment at hand was incapable of measuring those moments.

The serious glider performance "hacker" will recognize that the moment data is needed for the following reasons:

1. Essential for stabilizer sizing.
2. Determines if flaps are feasible for a new or old design (flaps greatly affect the moments generated by the wing).
3. The moments influence stabilizer loading, and therefore the stabilizer induced drag - up to ten percent of total drag or more.

Any one of the above reasons could give us the feeling of being left out in the cold. However, recognizing that performance estimation is nothing better than an educated guess, it is possible to loosen up the standards of computation and estimate that moment, too. Perhaps the best way to estimate the section moment coefficient is through the application of thin airfoil theory.



$$Z = \frac{\text{MEAN LINE HEIGHT}}{C} \qquad X = \frac{\text{DISTANCE FROM LE}}{C}$$



$$C_L = \frac{L}{qS} \qquad C_D = \frac{D}{qS} \qquad C_{mac} = \frac{M}{qSc}$$

$$q = \frac{1}{2}\rho V^2$$

ρ = air density

S = wing area

Figure 1

A complete discourse on thin airfoil theory is completely outside the scope of this short article. Basically, the airfoil can be mathematically modelled by an equivalent airfoil of zero thickness which is identical to the mean line of the airfoil in question (the mean line is midway between the top and bottom of the airfoil. Air flow and lift about the airfoil can be simulated by placing a continuous or discrete vortex distribution along the mean line and integrating the airfoil properties using Fourier Series techniques. Since that is really more technical than most technical masochists want to get, I will present the results only, along with the definitions of two functions that will be usefull shortly. For the sake of completeness, Figure 1 illustrates the lift, drag, and moment conventions on the airfoil, as well as the airfoil coordinates and the formulas for normalized coefficients.

$$\alpha_{0L.FT} = \int_0^1 z \frac{1}{(1-X)\sqrt{X-X^2}}\,dx \qquad \text{EQN. 1}$$

$$C_{MAC} = \frac{\pi}{2}\alpha_0 + 2\int_0^1 z \frac{1-2X}{\sqrt{X-X^2}}\,dx \qquad \text{EQN. 2}$$

$$f_1(x) = -\frac{1}{\pi}\left(\frac{1}{(1-X)\sqrt{X-X^2}}\right) \qquad \text{EQN. 3}$$

$$f_2(x) = \frac{1-2X}{\sqrt{X-X^2}} \qquad \text{EQN. 4}$$

It is time to specify what is being done here, so there will not be too much postage wasted on nasty letters. The moment coefficient of an airfoil moving through an incompressible and inviscid fluid (no Mach number or boundary layer effects) is constant as measured about the quarter-chord point, no matter what the angle of attack or lift coefficient. This point (constant moment) is the defined point for aerodynamic center of the section. If a weighted average of an entire wing is taken with respect to chord, the average quarter-chord location will also be close to the aerodynamic center of the wing. The problem, the single biggest problem for anyone concerned with model aircraft, is viscosity and effects associated with Reynolds number. The various kinds of boundary layer separation that occur on our airfoils throughout the entire flight regime prevent us from applying the abstract fluid theory to our real world with certainty. In practice, there is no aerodynamic center, and experimental results show a variation of quarter-chord moment coefficient with angle of attack. Usually we can apply the quarter chord moment coefficient predicted by the thin airfoil theory and get a fair estimate of the actual moments generated. However, it must be realized that large separation will entail large errors. Airfoils that exhibit highly nonlinear lift curves are particularly suspicious.

Keeping this in mind, we are ready to apply the theory. While the previous formulas are difficult to apply to an arbitrary airfoil in their raw form (eqns. 1 and 2), two functions may be tabulated to help smooth the calculation (eqns. 3 and 4). A rough estimate of zero-lift angle and moment coefficient may be performed with the aid of this table assembled by H. Glauert:

x	$f_1(x)$	$f_2(x)$
0.025	-2.09	6.10
0.05	-1.54	4.13
0.10	-1.18	2.67
0.20	-1.00	1.50
0.30	-0.99	0.87
0.40	-1.08	0.41
0.50	-1.27	0.00
0.60	-1.62	-0.41
0.70	-2.32	-0.87
0.80	-3.98	-1.50
0.90	-10.6	-2.67
0.95	-29.2	-4.13

Although it can be recognized that the values of these functions are zero at x=0, there is some significant calculus needed to learn that the value of $f_1(x)$ is zero at x=1.0, and highly dependent on the mean line shape between x=0.95 and x=1.0. In most cases the mean line can be assumed straight in this area, and substitution and integration over the interval will show that the contribution for $f_1(x)$ is -2.87 times the height of the mean line at x=0.95. The function $f_2(x)$ does not suffer from such wild behavior, though, and may be integrated over the whole chord using trapezoidal integration. For those unfamiliar with trapezoidal integration, an easy approximation to the integrals of equations 1 and 2 follows:

COORDINATES: $(x_1, z_1), (x_2, z_2), \ldots \quad x_1, z_1 = 0$

$x_N = 1.0, \; z_N = 0$

$$\int_0^1 z \cdot f(x)\,dx = \frac{z_1 f(x_1) + z_2 f(x_2)}{2}(x_2 - x_1) + \frac{z_2 f(x_2) + z_3 f(x_3)}{2}(x_3 - x_2) + \cdots + \frac{z_{N-1} f(x_{N-1}) + z_N f(x_N)}{2}(x_N - x_{N-1})$$

After going to all this trouble, what do we have? Well, the equations furnish the zero lift angle in radians, and we would probably desire that in degree units (multiply by 57.3). Now we know everything there is to know about the airfoil in inviscid, incompressible flow, and have a good idea of what it does in the real world. Naturally, there are ways to check the accuracy of this method. I have implemented a computer program which helps compute moment coefficients for arbitrary airfoils, in the BASIC language (see listing at the end of the article). Although I was unable as of this writing to check airfoil data that is exactly in our Reynolds number range, the basic computation should be valid for our estimation since viscous effects do not enter in to the method. The findings are summarized in the following table:

Airfoil	Computed C_{mac}	Actual C_{mac}
NACA 2412	-0.0446	-0.045
Eppler 662, 15% thk.	-0.16	-0.15
NACA a=0.3 7.2% mean line	-0.0943	-0.106

The table seems to show a good agreement with the real world.

Naturally, I have managed to keep the best part for last so that you, the reader, would have to read clear through this boring dissertation by me, the author. I have been able to run several airfoil coordinates through the program with the following results:

Airfoil	C_{mac}
FX60-100	-0.122
FX60-126	-0.118
FX63-137	-0.2367
FX-M2	-0.114
Eppler 193	-0.0769
NACA 2412	-0.045 *
GOE 795	-0.0573

* Note: NACA data is actual C_{mac} measured at higher Reynolds number.

I hope to compute the moment coefficients of several more airfoils as time allows, and send them along. In the meantime, please apply the following program, or your favorite calculator.

References

1. Abbott and von Doenhoff, Theory of Wing Sections

2. Kuethe and Schetzer, Foundations of Aerodynamics

AIRFOIL CMAC PROGRAM NOTES

This program is designed to be run on a wide variety of computers and only has a few features peculiar to its native machine, the Commodore VIC-20. Since Commodore uses an enhanced version of Microsoft BASIC, any machine that has a dialect of Microsoft BASIC or runs a BASIC close to the ANSI standard can run this program with few modifications. This includes all other Commodore computers, Northstar, the TRS-80 computers (except their Pocket Computer), the Apple II (especially easy for Applesoft), the IBM Personal Computer and their mainframe BASIC, the Wang computers, the Superbrain, and many other machines. The non-standard features are the PRINT#4 statements, which direct output to the printer, and the use of the pi symbol to represent the actual value of pi (3.14159...). I also used multiple statements on some line numbers, this is not allowed with a few interpreters, and on the Northstar the colon statement delimiter is replaced by a backslash. Rather than detail how to format the print for non-compatible machines, I will just remark that this is a fairly simple task which is made easier by examining the flow of the program, which follows.

AIRFOIL CMAC PROGRAM FLOW

INDEX	LINE #	OPERATION
		MAIN PROGRAM
1	160	Dimension arrays - DIM U(2,60),L(2,60)
2	165	Define integral functions - FNA(Z)=f_2(z) , FNB(Z)=f_1(z)
3	170	Move READ pointer to start of DATA statements - RESTORE
4	180	Read airfoil name, zero lift angle, coordinate % flag - N$,A,FC
5	190	Read upper coordinates - U(I,J) : I=1 is x coord. I=2 is z coord. J=1 to the # of coords.
6	Rea	Read lower coordinates - L(I,J) : similar to U(I,J)
7	220	Print airfoil name, header for coordinates
8	280	Print coordinates
9	360	Convert zero lift angle from degreees to radians

AIRFOIL CMAC PROGRAM FLOW (CONT'D)

INDEX	LINE #	OPERATION
		Main Program (Cont'd)
10	365	If coordinates are chord-normal go to step 12 (FC=0)
11	370	Convert coordinates from % to chord-normal
12	390	Initialize moment integral (IN=0), set to first coordinate point (XC=0.01)
13	*	Find upper/lower surface coordinates - GOSUB 500
14	395	Compute moment integral (IN), zero lift angle integral (IA) over interval x=0 to x=0.01
15	*	Find surface coordinates at x=0.99 - GOSUB 500
16	410	Compute moment and zero lift angle integral over the interval x=0.99 to x=1.00
17	420	Sum the moment and zero lift angle integrals from x=0.01 to x=0.99 with step delta x-0.01
18	460	Compute C_m with computed zero lift angle (C1) and C_m with given zero lift angle (C2)
19	470	Print moment coefficients and zero lift angles
END		
		Surface Coordinate Subroutine
1	510	Search for coordinate U(1,J) greater than XC
2	540	Compute upper surface coordinate by linear interpolation
3	560	Search for coordinate L(1,J) greater than XC
4	590	Compute lower surface coordinate by linear interpolation
5	610	Compute mean line coordinate (ZM)
RETURN		

Some notes of interest about the program:

The program will calculate the zero lift angle of the airfoil, input of zero lift angle from experiment is an effort to compensate for separation effects. The program uses a finer integration step than given by the function table, a delta x of 0.01. In the program, all "PRINT#4" statements are to the printer, simple "PRINT" to the CRT. The pi symbol translates to the value of pi on my machine.

```
10 REM AIRFOIL CMAC PROGRAM
20 REM
30 REM WALTER LOUNSBERY
40 REM 739 LITCHFIELD
50 REM WICHITA, KS  67203
60 REM
70 REM ***********
80 REM INSERT DATA STATEMENTS FOR AIRFOIL AT END OF PROGRAM
90 REM IN THIS ORDER:
100 REM "AIRFOIL NAME", ANGULAR DIFF. CHORDLINE TO ZERO LIFT LINE (DEGREES),
105 REM   COORD. % FLAG (1 = % OF CHORD, 0 = RATIO TO CHORD)
110 REM UPPER COORD. PAIRS AS -
115 REM   CHORDLINE DIST., SURFACE DIST. ABOVE CHORDLINE
120 REM LOWER COORD. PAIRS AS FOR UPPER SURFACE
130 REM *** ENTER COORDINATES FROM LE TO TE
140 REM WHEN DATA IS CHECKED AND READY, JUST RUN PROGRAM
150 REM
160 DIM U(2,60),L(2,60):OPEN4,4
165 DEF FNA(X)=X*(1-2*XC)/SQR(XC-XC*XC)
166 DEF FNB(X)=-1/((1-X)*SQR(X-X*X))
170 RESTORE
180 READ N$,A,FC
190 FOR J=1 TO 60: FOR I=1 TO 2:READ U(I,J):NEXT I:IF FC=1 THEN 193
191 IF U(1,J)=1.0 THEN 200
192 GO TO 195
193 IF U(1,J)=100 THEN 200
195 NEXT J
200 UN=J
210 FOR J=1 TO 60:FOR I=1 TO 2:READ L(I,J):NEXT I:IF FC=1 THEN 213
211 IF L(1,J)=1.0 THEN 216
212 GO TO 215
213 IF L(1,J)=100 THEN 216
215 NEXT J
216 LN=J
220 PRINT#4:PRINT#4,">>>>> ZERO LIFT MOMENT CALCULATION <<<<<"
230 PRINT#4:PRINT#4:PRINT#4,"AIRFOIL: ";N$
240 PRINT#4:PRINT#4,"COORDINATES"
250 PRINT#4:PRINT#4,"*** UPPER ***";SPC(12);"*** LOWER ***"
260 PRINT#4,"X/C";SPC(11);"Z/C";SPC(11);"X/C";SPC(11);"Z/C  %":PRINT#4
280 D=UN:IF LN>UN THEN D=LN
290 FOR J=1 TO D:PRINT#4," ";
300 IF J>UN THEN 320
305 U1$=STR$(U(1,J)):U2$=STR$(U(2,J)):D1=14-LEN(U1$):D2=14-LEN(U2$)
310 PRINT#4,U1$;SPC(D1);U2$;SPC(D2);
315 GO TO 330
320 PRINT#4,TAB(28);
330 IF J>LN THEN 350
340 L1$=STR$(L(1,J)):L2$=STR$(L(2,J)):D1=14-LEN(L1$):D2=14-LEN(L2$)
341 PRINT#4,L1$;SPC(D1);L2$;SPC(D2);
350 PRINT#4," ":NEXT J
360 A=A*π/180
365 IF FC=0 THEN 390
370 FOR J=1 TO UN:FOR I=1 TO 2:U(I,J)=U(I,J)/100:NEXT I,J
380 FOR J=1 TO LN:FOR I=1 TO 2:L(I,J)=L(I,J)/100:NEXT I,J
390 IN=0:XC=0.01:GOSUB 500
395 IA=FNB(XC)*ZM
400 IN=FNA(ZM):XC=0.99:GOSUB 500
410 IN=(IN+FNA(ZM))*0.005:IA=(IA+ZM*FNB(XC))*0.005
420 FOR XC=0.01 TO 0.990001 STEP 0.01
430 GOSUB 500
```

2

```
440 IN=IN+FNA(ZM)*0.01:IA=IA+FNB(XC)*ZM*0.01
450 NEXT XC
460 C1=2*IN+IA/2:C2=2*IN-π*A/2
470 PRINT#4:PRINT#4,"ZERO LIFT MOMENT COEFFICIENT:"
475 PRINT#4:PRINT#4,"    VIA COMPUTED ZERO LIFT ANGLE = ";C1
476 PRINT#4:PRINT#4,"    VIA ACTUAL ZERO LIFT ANGLE = ";C2
477 PRINT#4:PRINT#4,"COMPUTED ZERO LIFT ANGLE IS ";IA*180/(π*π);" DEGREES"
478 PRINT#4:PRINT#4,"ACTUAL ZERO LIFT ANGLE IS ";-A*180/π;" DEGREES"
480 CLOSE 4:END
500 REM MEANLINE SUBROUTINE
510 FOR J=2 TO UN
520 IF XC<=U(1,J) THEN 540
530 NEXT J
540 D=(XC-U(1,J-1))/(U(1,J)-U(1,J-1))
550 UC=D*(U(2,J)-U(2,J-1))+U(2,J-1)
560 FOR J=2 TO LN
570 IF XC<=L(1,J) THEN 590
580 NEXT J
590 D=(XC-L(1,J-1))/(L(1,J)-L(1,J-1))
600 LC=D*(L(2,J)-L(2,J-1))+L(2,J-1)
610 ZM=(LC+UC)/2
620 RETURN
700 DATA "WORTMANN FX 60-126",3,1
710 DATA 0,0
711 DATA .107,.675
712 DATA .428,1.349
713 DATA .961,2.096
714 DATA 1.704,2.802
715 DATA 2.653,3.493
716 DATA 3.806,4.174
717 DATA 5.156,4.808
718 DATA 6.699,5.457
719 DATA 8.427,6.021
720 DATA 10.332,6.585
721 DATA 12.408,7.077
722 DATA 14.645,7.555
723 DATA 17.033,7.958
724 DATA 19.562,8.327
725 DATA 22.221,8.615
726 DATA 25,8.859
727 DATA 27.886,9.019
728 DATA 30.866,9.13
729 DATA 33.928,9.16
730 DATA 37.059,9.138
740 DATA 40.245,9.041
741 DATA 43.474,8.893
742 DATA 46.73,8.679
743 DATA 50,8.425
744 DATA 53.27,8.118
745 DATA 56.526,7.781
746 DATA 59.755,7.402
747 DATA 62.941,6.994
748 DATA 66.072,6.549
749 DATA 69.134,6.082
750 DATA 72.114,5.589
751 DATA 75,5.084
752 DATA 77.779,4.567
753 DATA 80.438,4.055
754 DATA 82.967,3.552
755 DATA 85.355,3.07
756 DATA 87.592,2.611
757 DATA 89.668,2.181
758 DATA 91.573,1.777
```

```
759 DATA 93.301,1.412
760 DATA 94.844,1.084
761 DATA 96.194,.798
762 DATA 97.347,.554
763 DATA 98.296,.353
764 DATA 99.039,.198
765 DATA 99.572,.088
767 DATA 99.893,.024
768 DATA 100,0
800 REM LOWER SURFACE
810 DATA 0,0
811 DATA .107,-.301
812 DATA .428,-.641
813 DATA .961,-1.012
814 DATA 1.704,-1.404
815 DATA 2.653,-1.792
816 DATA 3.806,-2.132
817 DATA 5.156,-2.482
818 DATA 6.699,-2.761
819 DATA 8.427,-3.045
820 DATA 10.332,-3.262
821 DATA 12.408,-3.465
822 DATA 14.645,-3.598
823 DATA 17.033,-3.707
824 DATA 19.562,-3.746
825 DATA 22.221,-3.751
826 DATA 25,-3.683
827 DATA 27.886,-3.574
828 DATA 30.866,-3.392
829 DATA 33.928,-3.167
830 DATA 37.059,-2.877
840 DATA 40.245,-2.553
841 DATA 43.474,-2.188
842 DATA 46.73,-1.814
843 DATA 50,-1.421
844 DATA 53.27,-1.036
845 DATA 56.526,-.653
846 DATA 59.755,-.298
847 DATA 62.941,.029
848 DATA 66.072,.307
849 DATA 69.134,.547
850 DATA 72.114,.741
851 DATA 75,.897
852 DATA 77.779,1.006
853 DATA 80.438,1.073
854 DATA 82.967,1.093
855 DATA 85.355,1.074
856 DATA 87.592,1.022
857 DATA 89.668,.944
858 DATA 91.573,.845
859 DATA 93.301,.732
860 DATA 94.844,.61
861 DATA 96.194,.483
862 DATA 97.347,.357
863 DATA 98.296,.239
864 DATA 99.039,.146
865 DATA 99.572,.068
867 DATA 99.893,.014
868 DATA 100,0
8100 ,0,.574,-.2,1.144,-.436,1.775,-.691,2.368,-.97,2.948,-1.247,
```

```
10 REM AIRFOIL CMAC PROGRAM
20 REM
30 REM WALTER LOUNSBERY
40 REM 739 LITCHFIELD
50 REM WICHITA, KS  67203
60 REM
70 REM **********
80 REM INSERT DATA STATEMENTS FOR AIRFOIL ON TAPE
90 REM IN THIS ORDER:
100 REM "AIRFOIL NAME",
105 REM   COORD. % FLAG (1 = % OF CHORD, 0 = RATIO TO CHORD)
110 REM UPPER COORD. PAIRS AS -
115 REM   CHORDLINE DIST., SURFACE DIST. ABOVE CHORDLINE
120 REM LOWER COORD. PAIRS AS FOR UPPER SURFACE
130 REM *** ENTER COORDINATES FROM LE TO TE
150 REM
160 DIM U(2,60),L(2,60)
165 DEF FNA(X)=X*(1-2*XC)/SQR(XC-XC*XC)
166 DEF FNB(X)=-1/((1-X)*SQR(X-X*X))
167 PRINT:PRINT:PRINT"AIRFOIL NAME?":INPUT N$
168 IF N$="N" THEN N$=""
169 PRINT:PRINT"ZERO LIFT ANGLE?":INPUT A
170 PRINT:PRINT:PRINT"READY TAPE, THEN HIT   ANY KEY TO LOAD AIR-  FOIL"
171 GET A$:IF A$="" THEN 171
172 OPEN1,1,0,N$
180 INPUT#1,N$:INPUT#1,FC
190 FOR J=1 TO 60: INPUT#1,U(1,J):INPUT#1,U(2,J):IF FC=1 THEN 193
191 IF U(1,J)=1.0 THEN 200
192 GO TO 195
193 IF U(1,J)=100 THEN 200
195 NEXT J
200 UN=J
210 FOR J=1 TO 60:INPUT#1,L(1,J):INPUT#1,L(2,J):IF FC=1 THEN 213
211 IF L(1,J)=1.0 THEN 216
212 GO TO 215
213 IF L(1,J)=100 THEN 216
215 NEXT J
216 LN=J:CLOSE1:OPEN4,4
220 PRINT#4:PRINT#4,">>>>> ZERO LIFT MOMENT CALCULATION <<<<<"
230 PRINT#4:PRINT#4:PRINT#4,"AIRFOIL: ";N$
240 PRINT#4:PRINT#4,"COORDINATES"
250 PRINT#4:PRINT#4,"*** UPPER ***";SPC(12);"*** LOWER ***"
260 PRINT#4,"X/C";SPC(11);"Z/C";SPC(11);"X/C";SPC(11);"Z/C  %":PRINT#4
280 D=UN:IF LN>UN THEN D=LN
290 FOR J=1 TO D:PRINT#4," ";
300 IF J>UN THEN 320
305 U1$=STR$(U(1,J)):U2$=STR$(U(2,J)):D1=14-LEN(U1$):D2=14-LEN(U2$)
310 PRINT#4,U1$;SPC(D1);U2$;SPC(D2);
315 GO TO 330
320 PRINT#4,TAB(28);
330 IF J>LN THEN 350
340 L1$=STR$(L(1,J)):L2$=STR$(L(2,J)):D1=14-LEN(L1$):D2=14-LEN(L2$)
341 PRINT#4,L1$;SPC(D1);L2$;SPC(D2);
350 PRINT#4," ":NEXT J
360 A=A*π/180
365 IF FC=0 THEN 390
370 FOR J=1 TO UN:FOR I=1 TO 2:U(I,J)=U(I,J)/100:NEXT I,J
380 FOR J=1 TO LN:FOR I=1 TO 2:L(I,J)=L(I,J)/100:NEXT I,J
390 IN=0:XC=0.01:GOSUB 500
395 IA=FNB(XC)*ZM
400 IN=FNA(ZM):XC=0.99:GOSUB 500
410 IN=(IN+FNA(ZM))*0.005:IA=(IA+ZM*FNB(XC))*0.005
```

5

```
420 FOR XC=0.01 TO 0.990001 STEP 0.01
430 GOSUB 500
440 IN=IN+FNA(ZM)*0.01:IA=IA+FNB(XC)*ZM*0.01
450 NEXT XC
460 C1=2*IN+IA/2:C2=2*IN-π*A/2
470 PRINT#4:PRINT#4,"ZERO LIFT MOMENT COEFFICIENT:"
475 PRINT#4:PRINT#4,"    VIA COMPUTED ZERO LIFT ANGLE = ";C1
476 PRINT#4:PRINT#4,"    VIA ACTUAL ZERO LIFT ANGLE = ";C2
477 PRINT#4:PRINT#4,"COMPUTED ZERO LIFT ANGLE IS ";IA*180/(π*π);" DEGREES"
478 PRINT#4:PRINT#4,"ACTUAL ZERO LIFT ANGLE IS ";-A*180/π;" DEGREES"
480 CLOSE 4:END
500 REM MEANLINE SUBROUTINE
510 FOR J=2 TO UN
520 IF XC<=U(1,J) THEN 540
530 NEXT J
540 D=(XC-U(1,J-1))/(U(1,J)-U(1,J-1))
550 UC=D*(U(2,J)-U(2,J-1))+U(2,J-1)
560 FOR J=2 TO LN
570 IF XC<=L(1,J) THEN 590
580 NEXT J
590 D=(XC-L(1,J-1))/(L(1,J)-L(1,J-1))
600 LC=D*(L(2,J)-L(2,J-1))+L(2,J-1)
610 ZM=(LC+UC)/2
620 RETURN
8100 ,0,.574,-.2,1.144,-.436,1.775,-.691,2.368,-.97,2.948,-1.247,

READY.
```

READY.

```
10 REM AIRFOIL COORDINATE FILER
20 REM
30 REM WALTER LOUNSBERY
40 REM 739 LITCHFIELD
50 REM WICHITA, KS  67203
60 REM
70 REM ***********
80 REM INSERT DATA STATEMENTS FOR AIRFOIL AT END OF PROGRAM
90 REM IN THIS ORDER:
100 REM "AIRFOIL NAME"
105 REM   COORD. % FLAG (1 = % OF CHORD, 0 = RATIO TO CHORD)
110 REM UPPER COORD. PAIRS AS -
115 REM   CHORDLINE DIST., SURFACE DIST. ABOVE CHORDLINE
120 REM LOWER COORD. PAIRS AS FOR UPPER SURFACE
130 REM *** ENTER COORDINATES FROM LE TO TE
140 REM WHEN DATA IS CHECKED AND READY, JUST RUN PROGRAM
150 REM
160 DIM U(2,60),L(2,60)
170 RESTORE
180 READ N$,FC
190 FOR J=1 TO 60: FOR I=1 TO 2:READ U(I,J):NEXT I:IF FC=1 THEN 193
191 IF U(1,J)=1.0 THEN 200
192 GO TO 195
193 IF U(1,J)=100 THEN 200
195 NEXT J
200 UN=J
210 FOR J=1 TO 60:FOR I=1 TO 2:READ L(I,J):NEXT I:IF FC=1 THEN 213
211 IF L(1,J)=1.0 THEN 216
212 GO TO 215
213 IF L(1,J)=100 THEN 216
215 NEXT J
216 LN=J
218 PRINT:PRINT:PRINT"CHECK COORDINATES?    (HARD COPY)"
220 INPUT A$:IF LEFT$(A$,1)<>"Y" THEN 360
225 OPEN4,4
230 PRINT#4:PRINT#4:PRINT#4,"AIRFOIL: ";N$
240 PRINT#4:PRINT#4,"COORDINATES"
250 PRINT#4:PRINT#4,"*** UPPER ***";SPC(12);"*** LOWER ***"
260 PRINT#4,"X/C";SPC(11);"Z/C";SPC(11);"X/C";SPC(11);"Z/C  %":PRINT#4
280 D=UN:IF LN>UN THEN D=LN
290 FOR J=1 TO D:PRINT#4," ";
300 IF J>UN THEN 320
305 U1$=STR$(U(1,J)):U2$=STR$(U(2,J)):D1=14-LEN(U1$):D2=14-LEN(U2$)
310 PRINT#4,U1$;SPC(D1);U2$;SPC(D2);
315 GO TO 330
320 PRINT#4,TAB(28);
330 IF J>LN THEN 350
340 L1$=STR$(L(1,J)):L2$=STR$(L(2,J)):D1=14-LEN(L1$):D2=14-LEN(L2$)
341 PRINT#4,L1$;SPC(D1);L2$;SPC(D2);
350 PRINT#4," ":NEXT J
355 CLOSE4
360 PRINT:PRINT:PRINT"READY TAPE, THEN HIT   ANY KEY TO START"
370 GET A$:IF A$=""THEN 370
380 OPEN1,1,1,N$
390 PRINT#1,N$:PRINT#1,FC
400 FOR J=1 TO UN:PRINT#1,U(1,J):PRINT#1,U(2,J):NEXT J
410 FOR J=1 TO LN:PRINT#1,L(1,J):PRINT#1,L(2,J):NEXT J
420 CLOSE1
430 END
```

```
700 DATA "WORTMANN FX 60-126",1
710 DATA 0,0,              .107,       ,       .428,      ,        .961,
711 DATA 1.704,     ,      2.653,      ,      3.806,      ,       5.156
,
712 DATA6.699,      ,      8.427,      ,     10.332,      ,      12.408
,
713 DATA14.645,     ,     17.033,      ,     19.562,      ,      22.221
,
714 DATA25,         ,     27.886,      ,     30.866,      ,      33.928
,
715 DATA37.059,     ,     40.245,      ,     43.474,      ,       46.73,
716 DATA50,         ,      53.27,      ,     56.526,      ,      59.755
,
717 DATA62.941,     ,     66.072,      ,     69.134,      ,      72.114
,
718 DATA75,         ,     77.779,      ,     80.438,      ,      82.967
,
719 DATA85.355,     ,     87.592,      ,     89.688,      ,      91.573
,
720 DATA93.301,     ,     94.844,      ,     96.194,      ,      97.347
,
721 DATA98.926,     ,     99.039,      ,     99.572,      ,      99.893
,
722 DATA100,0
800 REM LOWER SURFACE
810 DATA 0,0,              .107,       ,       .428,      ,        .961,
811 DATA 1.704,     ,      2.653,      ,      3.806,      ,       5.156
,
812 DATA6.699,      ,      8.427,      ,     10.332,      ,      12.408
,
813 DATA14.645,     ,     17.033,      ,     19.562,      ,      22.221
,
814 DATA25,         ,     27.886,      ,     30.866,      ,      33.928
,
815 DATA37.059,     ,     40.245,      ,     43.474,      ,       46.73,
816 DATA50,         ,      53.27,      ,     56.526,      ,      59.755
,
817 DATA62.941,     ,     66.072,      ,     69.134,      ,      72.114
,
818 DATA75,         ,     77.779,      ,     80.438,      ,      82.967
,
819 DATA85.355,     ,     87.592,      ,     89.688,      ,      91.573
,
820 DATA93.301,     ,     94.844,      ,     96.194,      ,      97.347
,
821 DATA98.926,     ,     99.039,      ,     99.572,      ,      99.893
,
822 DATA100,0

READY.
```

>>>>> ZERO LIFT MOMENT CALCULATION <<<<<

AIRFOIL: WORTMANN FX 60-126

COORDINATES

*** UPPER ***		*** LOWER ***	
X/C	Z/C	X/C	Z/C %
0	0	0	0
.107	.675	.107	-.301
.428	1.349	.428	-.641
.961	2.096	.961	-1.012
1.704	2.802	1.704	-1.404
2.653	3.493	2.653	-1.792
3.806	4.174	3.806	-2.132
5.156	4.808	5.156	-2.482
6.699	5.457	6.699	-2.761
8.427	6.021	8.427	-3.045
10.332	6.585	10.332	-3.262
12.408	7.077	12.408	-3.465
14.645	7.555	14.645	-3.598
17.033	7.958	17.033	-3.707
19.562	8.327	19.562	-3.746
22.221	8.615	22.221	-3.751
25	8.859	25	-3.683
27.886	9.019	27.886	-3.574
30.866	9.13	30.866	-3.392
33.928	9.16	33.928	-3.167
37.059	9.138	37.059	-2.877
40.245	9.041	40.245	-2.553
43.474	8.893	43.474	-2.188
46.73	8.679	46.73	-1.814
50	8.425	50	-1.421
53.27	8.118	53.27	-1.036
56.526	7.781	56.526	-.653
59.755	7.402	59.755	-.298
62.941	6.994	62.941	.029
66.072	6.549	66.072	.307
69.134	6.082	69.134	.547
72.114	5.589	72.114	.741
75	5.084	75	.897
77.779	4.567	77.779	1.006
80.438	4.055	80.438	1.073
82.967	3.552	82.967	1.093
85.355	3.07	85.355	1.074
87.592	2.611	87.592	1.022
89.668	2.181	89.668	.944
91.573	1.777	91.573	.845
93.301	1.412	93.301	.732
94.844	1.084	94.844	.61
96.194	.798	96.194	.483
97.347	.554	97.347	.357
98.296	.353	98.296	.239
99.039	.198	99.039	.146
99.572	.088	99.572	.068
99.893	.024	99.893	.014
100	0	100	0

ZERO LIFT MOMENT COEFFICIENT:

VIA COMPUTED ZERO LIFT ANGLE = -.118105598

VIA ACTUAL ZERO LIFT ANGLE = -.0807641549 ?

COMPUTED ZERO LIFT ANGLE IS -4.36205253 DEGREES

ACTUAL ZERO LIFT ANGLE IS -3 DEGREES